Slide 1	**Differential Equation Models Using NM-TRAN** Nick Holford Dept Pharmacology & Clinical Pharmacology University of Auckland, New Zealand	The principles of using differential equations to define pharmacokinetic and pharmacodynamic models can be applied to many modelling programs. The use of NM-TRAN to code models for NONMEM illustrates some of these general principles and also involves some techniques that are peculiar to the NONMEM system.
Slide 2	Why Use DEQs? • Simpler to write and understand • Some models don't have closed form solution e.g. – Mixed Order elimination ("Michaelis-Menten) – Turnover models ("Indirect Effect PKPD") ©NHG Holford, 2015, all rights reserved.	In general it is easier to write a pharmacokinetic model using differential equations (DEQs). The time course of changes in biological systems can usually be thought of in terms of physical processes reflecting input and output. Differential equations provide a simple way of describing concentrations and effects by directly translating a biological process into a simple equation. There are some classes of pharmacokinetic model that cannot be expressed as closed form (or analytical) solution. However, these models can used by numerical solutions to a set of differential equations.
Slide 3	Terminology • Closed form or Analytical solution C(t)=Dose/V*exp(-CL/V*t) • Differential Equation dC/dt=-CL/V*C • The Solution C(t) • The Derivative -CL/V*C • Initial Condition C(0)=Dose/V ©NHG Holford, 2015, all rights reserved.	Terminology is important when describing how to use differential equations.

Slide 4

Closed Form and DEQ

- Closed Form

```
$SUB ADVAN1 TRANS2

$PK
CL=THETA(1)
V=THETA(2)
S1=V


$ERROR
Y=F+ERR(1)
```

- Differential Equation

```
$SUB ADVAN6 TOL=3
$MODEL COMP (CENTRAL)
$PK
CL=THETA(1)
V=THETA(2)
S1=V
$DES
DADT(1)=-CL*A(1)/V
$ERROR
Y=F+ERR(1)
```

©NHG Holford, 2015, all rights reserved.

This example shows equivalent NM-TRAN code for closed form and differential equation defined methods. The model is a one compartment first order elimination disposition model. The solution of the equation is provided in the $ERROR block using the default prediction variable F.

Slide 5

$SUB

- ADVAN Subroutines
  - ADVAN 6
    - Simplest DEQ ADVAN
  - ADVAN 8
    - Stiff DEQ systems
  - ADVAN 9
    - More robust than ADVAN 6
    - $AES [useful?]
  - ADVAN 13
    - NM7 Only
    - Faster with newer estimation methods?
- TOL option
  - 3 is usually OK,
  - 9 recommended for NM7 with NSIG=3, SIGL=9

©NHG Holford, 2015, all rights reserved.

The $SUBROUTINE record is used to identify the method used to solve differential equaitons. NONMEM provides three methods for solving differential equations.

The fastest and usually the most effective is ADVAN6. This uses a variable step size Runge-Kutte integrator.

When there is a big difference in the time constants ('half-lives') among the differential equations then the system is said to be stiff. Stiff systems can take a very long time to solve using Runge-Kutte methods (or may not be solvable ever). The ADVAN8 method uses a Gear integrator designed for solving stiff systems.

For non-stiff systems an alternative method is ADVAN9. It uses the Adams predictor-corrector method and may work better than ADVAN6 for some systems. It can be hard to predict which integration method is best for a particular problem. The best method can be identified by trying each method in turn setting MAXEVAL=0 and timing how long it takes to complete.

All the integrators use a common parameter to control the local precision of the solution. This is specified as the TOL option on the $SUBROUTINE record. Bigger numbers mean the integrator will attempt to find a higher precision solution. This can take longer and sometimes cause NONMEM to fail with an error message that TOL might be too large.

Slide 6

$MODEL

- COMP defines each compartment
- Names for compartments have no meaning
- Attributes:
  - DEFDOSE (Default for CMT data item)
  - DEFOBS (The "F" compartment)

©NHG Holford, 2015, all rights reserved.

The $MODEL record is used to define the number of compartments in the differential equation system. There are certain attributes to a compartment that sometimes need to be changed.
DEFDOSE defines the default dosing compartment if this is not specified in the NONMEM data item CMT.
DEFOBS defines the default observation compartment if this is not specified in the NONMEM data item CMT or PCMT.
It is usually easier and clearer to explicitly define the CMT item in the data set and to compute the observation prediction in $ERROR.

Slide 7

$DES

- Derivatives in DADT(1…NCOMP)
- Solutions in A(1…NCOMP)
- NONMEM takes care of all compartment input

This is OK

```
C1=A(1)/V
DADT(1) = -CL*C1
```

This is NOT OK

```
C1=A(1)/V
DADT(1) = RATE-CL*C1
```

©NHG Holford, 2015, all rights reserved.

The $DES record is used to compute quantities needed to solve the differential equations. Each compartment defined in the $MODEL record should have a differential equation defined in $DES.
The value of the differential equation is assigned to an element of the DADT array.
The solution of each differential equation is supplied as an element of the A array.
The order of the compartments in $MODEL corresponds to the array index in DADT and A.

NONMEM takes care of delivering drug to the compartment. You should not attempt to write a differential equation involving the input process that is already defined by the data file AMT and RATE data items.

Slide 8

$ERROR

- Can refer to solution as F or A()

```
Y=F + ERR(1)
Y=A(1)/V + ERR(1)
```

(This works for any ADVAN)

- Cannot use same variable name as $DES

```
$DES
   C1=A(1)/V
$ERROR
   CP=A(1)/V
```

not

```
C1=A(1)/V
```

©NHG Holford, 2015, all rights reserved.

The $ERROR record obtains the solution of the differential equation either in the default variable F or by directly computing it from the solution array A.

Be careful about variable names in $DES and $ERROR. It is often tempting to use the same name for a variable but NM-TRAN will identify this as an error.

Slide 9

CMT Data Item

- `CMT` links `AMT` to the model compartment
- Not specific to use of `$DES` but often needed with `$DES`
  - `AMT` are for compartment 1

#ID	TIME	AMT	CMT
1	0	100	1

  - `AMT` are for compartment 3

#ID	TIME	AMT	CMT
1	0	100	3

©NHG Holford, 2015, all rights reserved.

The CMT data item is commonly used when writing differential equation models. Unfortunately the design of NONMEM requires that this model specific value has to specified in the data set. This can be a nuisance because it means that changing your model requires a change to the data.

Slide 10

Initial Conditions

- Models may need to initialize compartments without a dose e.g. turnover models

```
dM/dt=Rin - Kout*M
At t=0:
  Rin=Kout*M(0)
```

©NHG Holford, 2015, all rights reserved.

When NONMEM was designed in the last millennium it was expected that $DES would only be used to model pharmacokinetic systems that always started with an initial value of zero for all the compartments. In NONMEM V there was no convenient way to specify the initial condition of a differential equation but all differential equations must have their initial condition defined!

If the initial condition is not zero e.g. when describing the kinetics of a physiological substance, a trick had be used to set the initial condition value. This is no longer required.

Slide 11

Initializing Compartments

- A_0FLG is set by NONMEM for the first record of each subject (i.e. NEWIND.LE.0)

(1) Explicit compartment initialization block:

```
IF (A_0FLG.EQ.1) THEN
 A_0(1)=THETA(1)*(1+ETA(1))
ENDIF
```

(2) Implicit compartment initialization block:

```
A_0(1)=THETA(1)*(1+ETA(1))
```

©NHG Holford, 2015, all rights reserved.

Slide 12

Steady State with I_SS

- Initial Steady State
  - Easier to initialize compartments for some special applications

I_SS= n
I_SS may be used with the general non-linear models (ADVAN6, ADVAN8, ADVAN9) to request the initial state feature of PREDPP. Values of n are
0 No initial state state (the default)
1 Initial steady state
2 Initial steady state, adds to current compartment amounts.
3 Initial steady state, use current compartment amounts as initial estimates.

©NHG Holford, 2015, all rights reserved.

Slide 13

TIME AND T

- `TIME` is fixed to the value on the current record
- `T` starts at `TIME` of previous record and advances to `TIME` of current record as the numerical integrator computes the solution

#ID	TIME	AMT	CMT	DV
1	0	100	1	0
1	10	.	.	20
1	30	.	.	10

```
$DES ; KA is function of time
KA=THETA(3)+THETA(4)*T
DADT(1)=KA*A(1) - CL*A(2)/V
```

©NHG Holford, 2015, all rights reserved.

It is important to understand the way that NONMEM uses the TIME data item when solving differential equations. The solution to the equation is obtained by integrating from the solution at the previous TIME value to the TIME of the current data record. The value of TIME is the value for the current data record e.g. when the equations are solved to provide a value at TIME=10 the integrator starts at 0 but the value of TIME will be 10.

The time used by the integrator is available in the $DES variable called T. It will start at 0 and end up at 10 in order to obtain a solution at TIME=10. Notice that only a few values of T are computed to obtain the solution. You cannot assume that T will have all possible values between 0 and 10. However, the values of T will be chosen by the integrator so that the solution will not depend on how T is used in the equation.

In the example it is assumed that KA is a function of time. The T variable is used to predict the time dependent value of KA.

Slide 14

Computation Efficiency

- Solution of differential equations is numerically demanding
- Avoid computation in $DES
- Pre-calculate fixed values in $PK

Inefficient

```
$PK
$DES
VMAX=THETA(1)*(WT/70)**0.75
KM=THETA(2)
V=THETA(3)*WT/70
DADT(1)= - VMAX*A(1)/V/(KM+A(1)/V)
```

Efficient

```
$PK
SIZE=WT/70
VMAX=THETA(1)*SIZE**0.75
KM=THETA(2)
V=THETA(3)*SIZE
$DES
C1=A(1)/V
DADT(1)= - VMAX*C1/(KM+C1)
```

©NHG Holford, 2015, all rights reserved.

The solution of differential equations involves a lot of calculations. Typically it takes about 10 times longer to solve a differential equation defined model compared with the closed form solution. It may help speed up NONMEM if you pay attention to the way that variables are computed. If a variable will not change during the solution of the equation then it should be computed in $PK.

Slide 15

DE Evaluations (NM VI)

```
$PK
; The following code allows the differential
; eqn solver to keep going when it is
; struggling to solve the equations

"FIRST
" COMMON /PRCOMG/ IDUM1,IDUM2,IMAX
" INTEGER IDUM1,IDUM2,IMAX
" IMAX=500000
```

IMAX is an internal NONMEM variable that limits the number of differential equation evaluations.
Verbatim code must be entered EXACTLY as shown

©NHG Holford, 2015, all rights reserved.

Sometimes NONMEM will complain that it cannot solve the DEs because too many differential equation evaluations are required.
Do not omit the leading blanks in records 2 to 4 of the verbatim code.
Default iteration maximum (IMAX) is 100000. Example shows how to change IMAX to 500000.

RE: [NMusers] help on a minimization problem
Ludden, Thomas
Tue, 22 Jun 2010 11:53:50 -0700
Ethan and Nick, To change the maximum number of calls during an integration interval for NM7 try placing the following at the beginning of $PK. "FIRST " USE PRCOM_INT, ONLY: IMAX and somewhere after all MU-referencing specify the IMAX value. For example, " IMAX=3000000 Beginning with version VI the default value is 1000000. To change the default value, redefine MAXFCN in SIZES and recompile.

Slide 16

DE Evaluations (NM 7)

To change the maximum number of calls during an integration interval for NM7 try placing the following at the beginning of $PK.

```
"FIRST
"     USE PRCOM_INT, ONLY: IMAX
```

and somewhere after all MU-referencing specify the IMAX value. For example,

```
" IMAX=3000000
```

IMAX is an internal NONMEM variable that limits the number of differential equation evaluations. Default is 1000000.
Verbatim code must be entered EXACTLY as shown

©NHG Holford, 2015, all rights reserved.

RE: [NMusers] help on a minimization problem
Ludden, Thomas
Tue, 22 Jun 2010 11:53:50 -0700
To change the maximum number of calls during an integration interval for NM7 try placing the following at the beginning of $PK.
"FIRST " USE PRCOM_INT, ONLY: IMAX
and somewhere after all MU-referencing specify the IMAX value. For example,
" IMAX=3000000
Beginning with version VI the default value is 1000000. To change the default value, redefine MAXFCN in SIZES and recompile.

Slide 17

References

Fehlberg E. Low-order classical Runge-Kutta formulas with stepsize control and their application to some heat transfer problems. NASA Technical Report 1969;R-315

Forsythe GE, Malcolm MA, Moler CB. Computer methods for mathematical computations. Englewood Cliffs: Prentice-Hall,Inc; 1977

©NHG Holford, 2015, all rights reserved.